

有限差分法による熱伝導方程式の数値計算

有限差分近似

- 与えられた関数 $f(x)$ の微分係数を任意の点 $x = a$ で数値的に計算したい。微分の定義式

$$\left(\frac{df}{dx}\right)_{x=a} = \lim_{\delta \rightarrow 0} \frac{f(a+\delta) - f(a)}{\delta}$$

において、 δ を有限にして近似しようというのが、以下に述べる有限差分法 (finite difference method) の基本的な考え方である。




図 1: 変数の離散化のイメージ図。

- まず計算機上で扱うことができるよう、連続変数を離散化する必要がある (図 1 を参照)。
 - 独立変数 x を離散化し、格子 (grid) をつくる。
 - i 番目の格子点 x_i における関数の値を $f_i \equiv f(x_i)$ で定義する。
 - 格子点 $x = x_i$ での $f(x)$ の微分を、その周囲の点での関数の値 f_j ($j = i, i \pm 1, \dots$) を用いて四則演算で近似する。
- 等間隔に格子が形成されていて、その隣り合う格子点どうしの間隔が h であるとする。まず $x = x_i$ なる格子点のまわりで関数 $f(x)$ をテーラー展開する。

$$f_{i+1} = f_i + h \left(\frac{df}{dx}\right)_{x=x_i} + \frac{1}{2}h^2 \left(\frac{d^2f}{dx^2}\right)_{x=x_i} + \frac{1}{6}h^3 \left(\frac{d^3f}{dx^3}\right)_{x=x_i} + \frac{1}{24}h^4 \left(\frac{d^4f}{dx^4}\right)_{x=x_i} + \dots \quad (1)$$

$$f_{i-1} = f_i - h \left(\frac{df}{dx}\right)_{x=x_i} + \frac{1}{2}h^2 \left(\frac{d^2f}{dx^2}\right)_{x=x_i} - \frac{1}{6}h^3 \left(\frac{d^3f}{dx^3}\right)_{x=x_i} + \frac{1}{24}h^4 \left(\frac{d^4f}{dx^4}\right)_{x=x_i} + \dots \quad (2)$$

これらを辺どうし足し算、引き算することで、1 階および 2 階微分係数の差分公式

$$\left(\frac{df}{dx}\right)_{x=x_i} = \frac{f_{i+1} - f_{i-1}}{2h} \quad (\text{誤差}) \simeq -\frac{1}{6}h^2 \left(\frac{d^3f}{dx^3}\right)_{x=x_i} \quad (3)$$

$$\left(\frac{d^2f}{dx^2}\right)_{x=x_i} = \frac{f_{i+1} - 2f_i + f_{i-1}}{h^2} \quad (\text{誤差}) \simeq -\frac{1}{12}h^2 \left(\frac{d^4f}{dx^4}\right)_{x=x_i} \quad (4)$$

が導かれる。これらは誤差が格子間隔 h の 2 乗に比例し、評価点での関数の値 f_i およびその両隣の値 $f_{i\pm 1}$ を使って微分係数を近似しているので、2 次の中心差分公式とか 3 点公式などという。

偏微分方程式の数値解法

- 時間 t と空間座標 (x, y, z) の関数である変数 $u(x, y, z, t)$ が、ある偏微分方程式を満たすとき、方程式にあらわれる空間偏微分を有限差分近似すれば、問題は、時間に関する連立常微分方程式を解くことに帰着する (これを「線の方法」という)。したがってこれまで学習した方法 (ルンゲ・クッタ法など) がそのままつかえる。
- 代表的な偏微分方程式を以下にかかげる。
 - 拡散方程式 (放物型の方程式)。物質の拡散、熱伝導によるエネルギー散逸、流体の粘性摩擦、磁気拡散など、散逸過程を記述する。下の式で $\eta > 0$ は拡散係数で、 $[\text{m}^2\text{s}^{-1}]$ の次元をもつ。系の大きさを L とすれば、特徴的な拡散時間は $\tau_\eta = L^2/\eta$ である。

$$\frac{\partial u}{\partial t} = \eta \left(\frac{\partial^2 u}{\partial x^2} + \frac{\partial^2 u}{\partial y^2} + \frac{\partial^2 u}{\partial z^2} \right) \equiv \eta \nabla^2 u \quad (5)$$

- 波動方程式 (双曲型の方程式)。波の伝搬をあらわす。 α は波の位相速度。左辺には時間に関する 2 階微分が含まれるので、補助変数を導入して、1 階の微分方程式に書き換えるとよい。

$$\frac{\partial^2 u}{\partial t^2} = \alpha^2 \left(\frac{\partial^2 u}{\partial x^2} + \frac{\partial^2 u}{\partial y^2} + \frac{\partial^2 u}{\partial z^2} \right) \quad (6)$$

- 拡散方程式や波動方程式を解くには、初期条件のほかに、境界条件 (boundary condition) が必要である。たとえば式 (5) の右辺には x の 2 階偏微分が含まれるので、境界条件を 2 個 (たとえば x 座標の定義域の端点においてそれぞれ 1 個ずつ) 与えなければならない¹。
- 境界において、変数の値そのものが与えられるとき、これをディリクレ (Dirichlet) 型の境界条件という。このとき境界では方程式を解く必要はなく、つねに与えられた値を代入しつづければよい。
- 境界 $x = a$ で、変数の偏微分係数の値が与えられるとき、これをノイマン (Neumann) 型の境界条件という。このとき境界でも方程式を解く必要がある。たとえば $x \geq a$ で変数が定義されていて、境界条件として

$$\left(\frac{\partial u}{\partial x} \right)_{x=a} = k \quad (7)$$

が与えられるとき、境界のすぐ内側の点 $x = a + h$ での変数の値は、テーラー展開公式 (1) を参照して、

$$u(a+h) = u(a) + kh + \frac{1}{2}h^2 \left(\frac{\partial^2 u}{\partial x^2} \right)_{x=a} + \frac{1}{6}h^3 \left(\frac{\partial^3 u}{\partial x^3} \right)_{x=a} + \dots, \quad (8)$$

と書けるから、 $x = a$ での u の 2 階微分係数に対する差分公式は、

$$\left(\frac{\partial^2 u}{\partial x^2} \right)_{x=a} = \frac{2[u(a+h) - u(a) - kh]}{h^2} \quad (\text{誤差}) \simeq -\frac{1}{3}h \left(\frac{\partial^3 u}{\partial x^3} \right)_{x=a} \quad (9)$$

のような 1 次の片側差分であらわすことができる。この結果は、仮に (本来存在しない) 境界のすぐ外側の点 $x = a - h$ にも変数が定義されているとあって、

$$u(a-h) \equiv u(a+h) - 2kh \quad \left(\text{あるいは} \frac{u(a+h) - u(a-h)}{2h} \equiv k \right) \quad (10)$$

を 2 次中心差分公式 (4) に代入したものと、形式的には一致する。一般に領域内部で中心差分 (3)(4) をつかっていれば、境界で 1 次精度の差分公式をつかっても、大域的には 2 次精度が保たれる。

演習課題 4 (6 月 20 日・21 日)

以下の問題 A, B を可能な限り解いて、これまでと同様、結果を電子メールで報告せよ (subject 名は “exercise 4”)。締め切りは 6 月 25 日 (月) とする。

問題 A

1 次元の熱伝導問題を考える。ある媒質の温度を、座標 x と時間 t の関数として $T(x, t)$ とおく (ただし $0 \leq x \leq a, t \geq 0$)。媒質の熱拡散率が一様に κ であるとき、温度場は偏微分方程式

$$\frac{\partial T}{\partial t} = \kappa \frac{\partial^2 T}{\partial x^2} \quad (11)$$

を満たす。問題設定は以下のとおり。まず媒質の初期温度は一様に T_0 であるとする。時刻ゼロの瞬間に、媒質の左端 ($x = 0$) の温度を急冷させ、 $T_0 - \Delta T$ に保つ。媒質の右端 ($x = a$) の境界条件は、

- 最初の温度 T_0 のまま保つ (等温壁)

¹一般には定義域の端点でなくてもよい。なんでもいから条件が合計 2 個あれば解ける。積分形で与えられることもある (大域的制約条件)。

- 温度勾配をゼロに保って熱流を遮断する (断熱壁)

の 2 通りを考える。

- (a) まず「等温壁」の場合を考える。長さを媒質の大きさ a , 時間を熱拡散時間 a^2/κ で規格化すると, 無次元の熱伝導方程式

$$\frac{\partial T}{\partial t} = \frac{\partial^2 T}{\partial x^2} \quad (12)$$

が導かれる。温度の単位を ΔT にとると, 初期条件は

$$T(x, 0) = 1, \quad (13)$$

境界条件は

$$T(0, t) = 0, \quad T(1, t) = 1 \quad (14)$$

と書くことができる。適当に格子を張り, 空間偏微分を 2 次の中心差分で置き換えると, 時間に関する連立常微分方程式が導かれる。これをオイラー法をもちいて時間積分せよ。安定に計算をすすめるには時間積分のステップ幅 Δ_t はある程度小さくしなければならない。どの程度小さくすればよいかを固有値解析によって議論し, それを実際に数値計算で確かめよ。安定な数値解が得られたら, 結果をわかりやすくグラフにして表現せよ。

- (b) つぎに「断熱壁」の場合を考える。方程式と初期条件とは前と同じだが, 境界条件は

$$T(0, t) = 0, \quad \left(\frac{\partial T}{\partial x}\right)_{x=1} = 0 \quad (15)$$

となる。同様に温度の時間発展を計算し, 結果をグラフにあらわせ。また領域の右端 $x = 1$ の温度が 0.5 になる時刻 τ_1 と, 領域中央の点 $x = 1/2$ における熱流束の大きさが最大になる時刻 τ_2 を 4 桁程度の精度で求めよ (Δ_t を小さくしたり, 格子点の個数を増やしたりしても, 結果が変わらないことを確認せよ)。

- (c) 陰解法である後退オイラー法またはクランク・ニコルソン法で時間積分を実行し, そのメリットについて議論せよ。

問題 B

2次元の熱伝導問題を考える。正方形の領域 $0 \leq x \leq 1, 0 \leq y \leq 1$ で無次元の熱伝導方程式

$$\frac{\partial T}{\partial t} = \frac{\partial^2 T}{\partial x^2} + \frac{\partial^2 T}{\partial y^2} \quad (16)$$

を, 初期条件 $T(x, y, t = 0) = 0$, 境界条件

$$T(x = 0, y, t) = 0, \quad T(x, y = 0, t) = 1, \quad \left(\frac{\partial T}{\partial x}\right)_{x=1} = \left(\frac{\partial T}{\partial y}\right)_{y=1} = 0, \quad (17)$$

のもとで解け (すなわち正方形の下辺 $y = 0$ と左辺 $x = 0$ は温度固定, 上辺 $y = 1$ と右辺 $x = 1$ は断熱境界)。じゅうぶん時間がたったあとの温度分布を, xy 平面にコンター図 (等温度線のプロット) としてあらわせ。またそのときの点 $(x, y) = (0.5, 1)$ での温度, および点 $(0, 1)$ での温度勾配の大きさを 4 桁程度の精度で求めよ。

解法のヒントと補足

- 「等温壁」の問題はディリクレ型の境界条件に相当する。境界では方程式を解く必要がない。たとえば格子を $x_i = i/N$ ($0 \leq i \leq N$) のように定義したとすれば, $1 \leq i \leq N - 1$ でのみ方程式を解けばよ

い。方程式 (12) にあられる空間 2 階微分を中心差分で近似すると、問題は時間に関する連立常微分方程式

$$\frac{d}{dt} \begin{pmatrix} T_0 \\ T_1 \\ \vdots \\ T_{N-1} \\ T_N \end{pmatrix} = \frac{1}{h^2} \begin{pmatrix} 0 \\ T_0 - 2T_1 + T_2 \\ \vdots \\ T_{N-2} - 2T_{N-1} + T_N \\ 0 \end{pmatrix} \quad (18)$$

を解くことに帰着する。ただし $T_i(t) = T(x_i, t)$ と定義した (境界条件より $T_0(t) = 0, T_N(t) = 1$)。変数の数が $N + 1$ 個であるから、やはり長さ $N + 1$ の配列 $w(0:n)$ (n は定数) を定義してそこに温度データを格納してやれば、あとは前回、前々回と同様である。

- 「断熱壁」ではノイマン型の境界条件が課せられる。1 次の片側差分 (9) を適用すると、解くべき連立常微分方程式は

$$\frac{d}{dt} \begin{pmatrix} T_0 \\ T_1 \\ \vdots \\ T_{N-1} \\ T_N \end{pmatrix} = \frac{1}{h^2} \begin{pmatrix} 0 \\ T_0 - 2T_1 + T_2 \\ \vdots \\ T_{N-2} - 2T_{N-1} + T_N \\ 2T_{N-1} - 2T_N \end{pmatrix} \quad (19)$$

となる。この場合 $T_N(t)$ も時間変化することに注意。

- いずれにしる問題は、線形の連立常微分方程式

$$\frac{d}{dt} \mathbf{w} = A\mathbf{w}, \quad \text{ただし } \mathbf{w}(t) = \begin{pmatrix} T_0(t) \\ \vdots \\ T_N(t) \end{pmatrix}, \quad (20)$$

を解くことに帰着する。係数行列 A (3 重対角行列のはずである) の $N + 1$ 個の固有値を λ_i ($0 \leq i \leq N$) とすれば、固有ベクトルを横に並べた行列 P をもちいて、行列 A は

$$P^{-1}AP = \begin{pmatrix} \lambda_0 & & \\ & \ddots & \\ & & \lambda_N \end{pmatrix} \quad (21)$$

のように対角化される。よって $\mathbf{w}' \equiv P^{-1}\mathbf{w}$ と定義すれば、問題は

$$\frac{dw'_i}{dt} = \lambda_i w'_i \quad (0 \leq i \leq N) \quad (22)$$

なる $N + 1$ 個の独立な 1 変数常微分方程式を解くことと等価である。あとは演習課題 1 と同様に、時間積分の安定性を議論することができる。

- オイラー法やルンゲ・クッタ法などの陽解法をとる限り、時間ステップの幅は

$$\Delta_t < O(h^2) \quad (23)$$

に制限される。いっぽう後退オイラー法やクランク・ニコルソン法などの陰解法は、固有値の実部が負であればつねに安定なので、 Δ_t に関する制限は事実上なくなり、少ない時間ステップで計算を終わらせることができる。たとえば後退オイラー法の場合、解くべき式は、ベクトルの形式 (20) で

$$\mathbf{w}^{(n+1)} = \mathbf{w}^{(n)} + \Delta_t A \mathbf{w}^{(n+1)}, \quad \text{すなわち } (I - \Delta_t A) \mathbf{w}^{(n+1)} = \mathbf{w}^{(n)}, \quad (24)$$

となる (ただし $\mathbf{w}^{(n)}$ は n 番目の時間ステップでの変数の値、 I は単位行列)。

- 3 重対角行列を係数行列にもつ連立一次方程式の解き方について。解くべき方程式を

$$Ax = r, \quad A = \begin{pmatrix} b_1 & c_1 & & & \\ a_2 & b_2 & c_2 & & \\ & \ddots & \ddots & \ddots & \\ & & & a_m & b_m \end{pmatrix}, \quad r = \begin{pmatrix} r_1 \\ r_2 \\ \vdots \\ r_m \end{pmatrix}, \quad x = \begin{pmatrix} x_1 \\ x_2 \\ \vdots \\ x_m \end{pmatrix},$$

と定義すれば, ガウスの消去法にもとづくプログラムはたとえば次のようである:

! 3 重対角 (tridiagonal) 行列を係数にもつ連立一次方程式を解く。まず行列変形のパート。

```

subroutine tridiag_init (a, b, c, m)
  implicit none
  integer(4) :: m, i
  real(8), dimension(1:m) :: a, b, c
  ! m は行列のサイズ
  ! 行列の係数
  do i = 1, m
    if (i == 1) then
      b(i) = 1.0d0 / b(i)
    else
      b(i) = 1.0d0 / (b(i) - a(i)*c(i-1))
      a(i) = b(i) * a(i)
    end if
    c(i) = b(i) * c(i)
  end do
  return
end subroutine tridiag_init
! つづいて求解のパート
subroutine tridiag (r, a, b, c, m)
  implicit none
  integer(4) :: m, i
  real(8), dimension(1:m) :: r, a, b, c
  ! r は入力でありかつ出力 (方程式の解)
  ! forward elimination
  r(1) = b(1) * r(1)
  do i = 2, m
    r(i) = b(i) * r(i) - a(i) * r(i-1)
  end do
  ! backward substitution
  do i = m-1, 1, -1
    r(i) = r(i) - c(i) * r(i+1)
  end do
  return
end subroutine tridiag
    
```

入力 r_i は方程式の右辺であるが, 同時にそれが求めるべき解 x_i に置き換わって出力される。いまの問題では行列 A は時間に依存しないので, まず時間積分の前に行列変形の部分 (tridiag_init) を 1 回だけ実行し, 積分開始後は求解の部分 (tridiag) を毎時間ステップ実行すればよい。

- クランク・ニコルソン法ならば解くべき式は,

$$\mathbf{w}^{(n+1)} = \mathbf{w}^{(n)} + \frac{1}{2}\Delta_t A (\mathbf{w}^{(n+1)} + \mathbf{w}^{(n)}), \quad \text{すなわち} \quad (I - \frac{1}{2}\Delta_t A)\mathbf{w}^{(n+1)} = (I + \frac{1}{2}\Delta_t A)\mathbf{w}^{(n)}, \quad (25)$$

である。行列 $I \pm \frac{1}{2}\Delta_t A$ はともに 3 重対角行列であるが, 行列

$$B \equiv (I - \frac{1}{2}\Delta_t A)^{-1}(I + \frac{1}{2}\Delta_t A) \quad (26)$$

は密行列 (すべての行列成分が非零) になってしまう。したがって行列 B を使って一気に計算してしまわずに, 2 つの行列 $I \pm \frac{1}{2}\Delta_t A$ に分けたままで, 2 段階に計算した方が, 必要となる計算量 (足し算, かけ算の回数) は減る。

データ処理についてのヒントと補足

- 出力ファイルの形式について。一定の時間間隔ごとに温度データを出力することになるだろうが,

- 1つのファイルにすべての時刻の温度データをまとめて書き出す (方法 1),
- 各時間ステップごとの温度データを別ファイルに書き出す (方法 2),

という2通りが考えられる。

• 方法1のプログラム例 (一部省略):

```
integer(4), parameter :: n = 10           ! 格子点の数は n+1
integer(4) :: i, it                       ! it は時間ステップをあらわす
real(8) :: t                              ! t は時刻
real(8), dimension(0:n) :: w             ! 温度データを格納する配列
real(8), parameter :: h = 1.0d0/n        ! 格子の間隔
it = 0                                    ! 時間ステップの初期化
t = 0.0d0                                 ! 時刻の初期化
do
  ! 100 時間ステップごとにデータを出力することにする
  if ( mod(it, 100) == 0 ) then
    write(*, "(' # t = ', e14.6)") t      ! 現在の時刻を出力しておく (先頭に # があれば
    do i = 0, n                          ! Gnuplot は無視する)
      write(*, "(2e16.8)") i*h, w(i)     ! 温度データを出力 (x座標, 温度の並び)
    end do
    write(*, *)                          ! ここで空行を出力するのがみそ
  end if
  ! 時間ステップをすすめる
  it = it + 1
  t = t + dt
end do
```

• 上の例のようにデータファイルの途中に空行を入れると Gnuplot でのグラフ作成に都合がよい。

- Gnuplot では空行は「データブロック」の区切りを意味し、とくに2次元データ (いまの例では時間と空間の2次元) のプロットに威力を発揮する。
- 出力結果をファイル temp.dat に保存したとすれば、単純に

```
gnuplot> plot 'temp.dat' with lines
```

によって、すべての時間ステップにおける温度分布が一枚のグラフにまとめて表示される。

- k 番目の時間ステップの温度分布だけをプロットするのも簡単である。

```
gnuplot> k = 5
gnuplot> plot 'temp.dat' every ::k::k with lines
```

- ソースプログラム中で、write 文によるデータ出力のところを

```
write(*, "(3e16.8)") t, i*h, w(i)
```

のように、時刻, x 座標, 温度の並びに変えてやれば、時空間上の3次元プロットも可能である。

```
gnuplot> splot 'temp.dat' with lines
```

• 方法2のプログラム例。ファイル名に連番の番号をつけてやるとあとでファイル操作が容易になる。

```
integer(4), parameter :: n = 10           ! 格子点の数は n+1
integer(4) :: i, it, ifile                ! it は時間ステップの番号, ifile はファイルの番号
real(8), dimension(0:n) :: w             ! 温度データを格納する配列
character(16) :: c                        ! ファイル名を代入するための文字列
character(8), parameter :: fn = 'temp'   ! ファイル名の先頭部分を8文字以内で任意に指定
real(8), parameter :: h = 1.0d0/n        ! 格子の間隔
ifile = 0
it = 0
do
  ! 100 時間ステップごとにデータを出力することにする。
  if ( mod(it, 100) == 0 ) then
    ! 変数 c に文字列を代入する。これでたとえば temp0000.dat などという文字列がつかれる。
    ! 組み込み関数 adjustl(文字列) は文字列の先頭の空白を除いて左寄せする。
    write(*, "(3e16.8)") t, i*h, w(i), c
```

```

! 同じく関数 trim(文字列) は末尾の空白を取り除く。
write(c, "(a, i4.4, '.dat')") trim(adjustl(fn)), ifile
open(10, file=trim(c)) ! ファイルをオープン
do i = 0, n
  write(10, "(2e14.6)") dble(i)*h, w(i) ! 温度データを出力 (x 座標, 温度の並び)
end do
close(10) ! ファイルをクローズ
ifile = ifile + 1 ! ファイル番号をインクリメント
end if
it = it + 1
end do

```

これで temp0000.dat, temp0001.dat, ..., というファイルができる。つぎのスクリプトはこれらをもとに、温度分布の絵を GIF 形式でたくさん描く。

```

#!/bin/sh
name=temp
i=0
while [ ${i} -lt 50 ]
do
  num=$(echo ${i} | awk '{printf("%04d", $1)}')
  gnuplot <<EOF
  set terminal gif
  set output '${name}${num}.gif'
  plot '${name}${num}.dat' with lines
  set output
EOF
  i=$((++i))
done

```

演習課題 2 の解答例

- (a) 省略 (配付資料の図 7 をみよ)。
- (b) 4 次ルンゲ・クッタ (RK4) 法は比較的良好にエネルギーを保存する。図 1(a) は横軸に時間、縦軸にエネルギーの相対誤差 $\varepsilon = |E(t)/E(0) - 1|$ をプロットしたものである。 $\varepsilon \ll 1$ の範囲では、誤差は経過時間 t に比例すること、そして Δ_t を半分にすると、ある時刻での誤差が $1/2^5 = 1/32$ になること (すなわち $\varepsilon \propto (\Delta_t)^5 t$ であること) がわかる。
- (c) 行列 A の固有値はすべて負の実数で、そのうち絶対値最大のもの λ_{\max} は、たとえば $N = 10$ のとき $\lambda_{\max} = -3.919$ であった。したがって運動方程式を 1 階の連立常微分方程式とみたときの固有値のうち絶対値最大のものは $\pm\sqrt{\lambda_{\max}} \simeq \pm 1.980i \equiv \pm i\omega_{\max}$ (i は虚数単位) で、純虚数となる。RK4 法の安定領域の虚軸上に着目すると、 $|\omega_{\max}\Delta_t| \leq 2\sqrt{2}$ となる必要があるから、時間積分の刻み幅は $\Delta_t \leq 2\sqrt{2}/\omega_{\max} \simeq 1.428$ と予想される。実際、図 2 に示されるように、 Δ_t がこの上限値をすこしでも上回ると、解が発散してしまうことがわかる。
- (d) オイラー法や 2 次ルンゲ・クッタ (RK2) 法では、安定領域に虚軸が含まれない。したがってこれらの方法では数値解の振幅はいずれ発散する。誤差の点でもこれらの方法は劣る。図 1(b) は RK2 法のエネルギー誤差を示しているが、たとえば $t = 1000$ でエネルギーの相対誤差を 1% 程度に抑えるためには、RK4 法では $\Delta_t = 0.1$ にすればすでにじゅうぶんであるのに対して、RK2 法では $\Delta_t = 0.025$ でもまだ足りない。RK4 法で時間ステップを 1 つ進めるのに要する計算時間は、RK2 法の場合にくらべて 2 倍程度であるが、それを加味しても、RK4 法の優位性はゆるがない。精度よく長時間積分するとき、RK4 法が適しているといえる。ただし精度はそれほど高くなくてもよく、また積分時間が短くていい場合は、計算時間の点からして、RK2 法など、より低次の方法で支障はない。
- (e) 運動のようすやエネルギー保存については省略。 $N = 10$ として、5 番目の質点の質量を $s = 1/10$ 倍、10 倍に変えた場合、行列 A の絶対値最大の固有値はそれぞれ $\lambda_{\max} = -21.052, -3.734$ になった。し

たがって Δ_t に関する制約条件は、それぞれ $\Delta_t \leq 0.616, 1.463$ と予想できる。系を安定に積分するために必要な Δ_t の値は、質点のなかでもっとも軽いもの（言い換えれば、質点の時定数を $\sqrt{m_j/k}$ で定義すると、もっとも時定数の短いもの）によって制約されるといえる。




図 1: $c_j = 1$ ($1 \leq j \leq N$), $N = 10$, $\alpha = 1/10$ のときの系の全エネルギーの相対誤差を時間の関数としてプロットした。(a) 4次ルンゲ・クッタ法の場合。(b) 改良オイラー法の場合。それぞれ 5 本の曲線が示されているが、それらは $\Delta_t = 0.4, 0.2, 0.1, 0.05, 0.025$ としたときの結果である。基本的に Δ_t を小さくすれば誤差は小さくなる。4次ルンゲ・クッタ法では、誤差は $(\Delta_t)^5 t$ に比例し、改良オイラー法では $(\Delta_t)^3 t$ に比例する。




図 2: $c_j = 1$ ($1 \leq j \leq N$), $N = 10$, $\alpha = 1/10$ のとき、5番目の質点の変位 $u_5(t)$ を時間の関数としてプロットしたものの。上からそれぞれ $\Delta_t = 1.43, 1.42, 0.01$ の結果。 $\Delta_t = 0.01$ の数値解はほぼ厳密解に近いと考えてよい。時間積分の刻み幅 Δ_t に対する安定条件が破られるやいなや、数値解は発散することがわかる。また安定条件ぎりぎりの Δ_t を与えると、厳密解に比べて、高周波成分がいちじるしく減衰してしまうことがわかる。